
* * * First Change * * * *
[bookmark: _Toc44847501][bookmark: _Toc51845155][bookmark: _Toc51845486][bookmark: _Toc51847006][bookmark: _Toc57022637]5.2.3.2.9	3gpp-Sbi-Oci
The header contains a comma-delimited list of Overload Control Information (OCI). See clause 6.4.3.
The encoding of the header follows the ABNF as defined in IETF RFC 7230 [12].

Sbi-Oci-Header = "3gpp-Sbi-Oci:" OWS 1#( timestamp ";" RWS validityPeriod ";" RWS olcMetric ";" RWS olcScope ) OWS

timestamp = "Timestamp:" RWS DQUOTE date-time DQUOTE

validityPeriod = "Period-of-Validity:" RWS 1*DIGIT "s"

olcMetric = "Overload-Reduction-Metric:" RWS ( DIGIT / %x31-39 DIGIT / "100" ) "%"

olcScope = nfProducerScope / nfConsumerScope / scpScope / seppScope

nfProducerScope = (   ( "NF-Instance:" RWS nfinst )
                    / ( "NF-Set:" RWS nfset )
                    / ( "NF-Service-Instance:" RWS nfservinst [ ";" RWS "NF-Inst:" RWS nfinst ] )
                    / ( "NF-Service-Set:" RWS nfserviceset )
                  ) [ ";" RWS sNssaiList ";" RWS dnnList ]

nfConsumerScope = ( "NF-Instance:" RWS nfinst [ ";" RWS "Service-Name:" RWS servname ] )
                / ( "NF-Set:" RWS nfset [ ";" RWS "Service-Name:" RWS servname ] )
                / ( "NF-Service-Instance:" RWS nfservinst [ ";" RWS "NF-Inst:" RWS nfinst ] )
                / ( "NF-Service-Set:" RWS nfserviceset )
                / ( "Callback-Uri:" RWS DQUOTE URI DQUOTE *( RWS "&" RWS DQUOTE URI DQUOTE ) )

scpScope =	"SCP-FQDN:" RWS fqdn

seppScope =	"SEPP-FQDN:" RWS fqdn

dnnList = "DNN:" RWS 1*tchar *( RWS "&" RWS 1*tchar )

sNssaiList = "S-NSSAI:" RWS snssai *( RWS "&" RWS snssai )

snssai = 1*tchar


"Timestamp" (Mandatory parameter): The date-time type is specified in IETF RFC 5322 [37] and clause 7.1.1.1 of IETF RFC 7231 [11]. It indicates the timestamp at which the overload control information was generated.
"Period-of-Validity" (Mandatory parameter): Period of validity is a timer that is measured in seconds. Once the timer expires, the OCI becomes invalid.
"Overload-Reduction-Metric" (Mandatory parameter): Overload-Reduction-Metric up to 3 digits long decimal string and the value range shall be from 0 to 100.
The Overload Control Scope is a mandatory header component, and it shall contain one of the parameters: "NF-Instance", "NF-Set", "NF-Service-Instance" or "NF-Service-Set" (for NF consumers or NF producers), "Callback-URI" (for NF consumers), "SCP-FQDN" (for SCP), or "SEPP-FQDN" (for SEPP).
See clause 6.4.3.4.5. The nfinst, nfset, nfservinst, nfserviceset and servname parameters are defined in clause 5.2.3.2.8. fqdn shall encode an FQDN. URI is defined in clause 3 of IETF RFC 3986 [14].
When signaling overload control information of an NF service instance, the "NF-Inst" parameter shall be present to identify the NF service instance unambiguously. If the "NF-Inst" parameter is absent, the receiving NF should assume the last known NF instance ID of NF service producer or consumer, if available.
NOTE 1:	Implementations complying with earlier versions of the specification can signal overload control information of an NF service instance without including the NF-Inst parameter.
"DNN" (Optional parameter): Used for S-NSSAI/DNN based overload control by SMF, see clause 6.4.3.4.5.2.2, that refers to one or more specific DNN(s). DNN format is defined in 3GPP TS 23.003 [15].
"S-NSSAI" (Optional parameter): Used for S-NSSAI/DNN based overload control by SMF, see clause 6.4.3.4.5.2.2, that refers to one or more specific S-NSSAI(s).
S-NSSAI format is defined in clause 5.4.4.2 of 3GPP TS 29.571 [13]. It shall be encoded as the object format (i.e. not converted to the string pattern defined in clause 5.4.4.2 of 3GPP TS 29.571 [13]).
EXAMPLE 1:	Overload Control Information for an NF Instance:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 75s; Overload-Reduction-Metric: 50%; NF-Instance: 54804518-4191-46b3-955c-ac631f953ed8
EXAMPLE 2:	Overload Control Information for an NF Service Set:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 120s; Overload-Reduction-Metric: 50%; NF-Service-Set: setxyz.snnsmf-pdusession.nfi54804518-4191-46b3-955c-ac631f953ed8.5gc.mnc012.mcc345
EXAMPLE 3:	Overload Control Information for an SMF instance related to a particular DNN of an S-NSSAI:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 600s; Overload-Reduction-Metric: 50%; NF-Instance: 54804518-4191-46b3-955c-ac631f953ed8; S-NSSAI: %7B%22sst%22%3A 1%2C %22sd%22%3A %22A08923%22%7D; DNN: internet.mnc012.mcc345.gprs
NOTE 2:	The S-NSSAI parameter corresponds to the JSON encoding: {"sst": 1, "sd": "A08923"} (see clause 5.2.3.1)
EXAMPLE 4:	Overload Control Information for an SMF instance related to a particular DNN shared by two S-NSSAIs:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 240s; Overload-Reduction-Metric: 50%; NF-Instance: 54804518-4191-46b3-955c-ac631f953ed8; S-NSSAI: %7B%22sst%22%3A 1%2C %22sd%22%3A %22A08923%22%7D & %7B%22sst%22%3A 1%2C %22sd%22%3A %22A08924%22%7D; DNN: internet.mnc012.mcc345.gprs
NOTE 3:	The S- NSSAI parameter corresponds to the JSON encoding: {"sst": 1, "sd": "A08923"} & {"sst": 1, "sd": "A08924"} (see clause 5.2.3.1)
EXAMPLE 5:	Overload Control Information sent by a NF service consumer with a scope set to a Callback-Uri:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 120s; Overload-Reduction-Metric: 25%; Callback-Uri: "https://pcf12.operator.com/serviceY"
EXAMPLE 6:	Overload Control Information sent by a NF service consumer with a scope set to a specific NF Instance and service:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 120s; Overload-Reduction-Metric: 25%; NF-Instance: 54804518-4191-46b3-955c-ac631f953ed8; Service-Name: nsmf-pdusession
EXAMPLE 7:	Overload Control Information sent by an SCP:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 120s; Overload-Reduction-Metric: 25%; SCP-FQDN: scp1.example.com
EXAMPLE 8:	Example with two OCI values, one for an SMF Instance and another one for a specific DNN of an S-NSSAI for the same SMF Instance:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 75s; Overload-Reduction-Metric: 50%; NF-Instance: 54804518-4191-46b3-955c-ac631f953ed8
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 600s; Overload-Reduction-Metric: 40%; NF-Instance: 54804518-4191-46b3-955c-ac631f953ed8; S-NSSAI: %7B%22sst%22%3A 1%2C %22sd%22%3A %22A08923%22%7D; DNN: internet.mnc012.mcc345.gprs
NOTE 4:	The S-NSSAI parameter corresponds to the JSON encoding: {"sst": 1, "sd": "A08923"} (see clause 5.2.3.1)
EXAMPLE 9:	Overload Control Information sent by an SEPP:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 120s; Overload-Reduction-Metric: 25%; SEPP-FQDN: sepp1.example.com
NOTE 5:	Example 8 is formatted as two distinct headers (which improves the readability), but it can also be formatted as a single header with two OCI values separated by a comma.
EXAMPLE 10:	Overload Control Information for an NF Service Instance:
3gpp-Sbi-Oci: Timestamp: "Tue, 04 Feb 2020 08:49:37 GMT"; Period-of-Validity: 75s; Overload-Reduction-Metric: 50%; NF-Service-Instance: xyz; NF-Inst: 54804518-4191-46b3-955c-ac631f953ed8

* * * Next Change * * * *
6.4.3.4.5.3	Scope of OCI signalled by an NF Service Consumer
The OCI sent by an NF Service Consumer shall include one and only one of the parameters defined in Table 6.4.3.4.5.3-1.
Table 6.4.3.4.5.3-1: Supported scopes for OCI signalled by an NF Service Consumer
	Parameter
	Value
	OCI scope (i.e. OCI applies to)
	Examples

	Callback-Uri
	One or more URI(s) including a scheme, authority and an optional path
	All notifications (or callbacks) with a notification (or callback) URI matching the Callback-Uri parameter value.

(NOTE 1)
	Callback-Uri:  "https://pcf12.operator.com"

Callback-Uri: "https://pcf12.operator.com/serviceY"

Callback-Uri:  "https://pcf12.operator.com/serviceY/abc" & "https://pcf12.operator.com/serviceY/def"


	NF-Instance
	NF Instance ID
	All notifications (or callbacks) bound to:
[bookmark: _PERM_MCCTEMPBM_CRPT57490069___7]-	the NF Instance ID; or
-	an NF service instance or an NF service set of this NF Instance ID.
(NOTE 2)
	NF-Instance: 54804518-4191-46b3-955c-ac631f953ed8

	NF-Set
	NF Set ID
	All notifications (or callbacks) bound to:
[bookmark: _PERM_MCCTEMPBM_CRPT57490070___7]-	the NF Set ID;
-	an NF instance of the NF Set ID; or
-	an NF service instance or an NF service set of an NF Instance of the NF Set ID.
(NOTE 2)
	NF-Set:  set1.udmset.5gc.mnc012.mcc345

	NF-Service-Instance (and NF-Inst)
	NF Service Instance ID (and NF Instance ID)
	All notifications (or callbacks) bound to:
[bookmark: _PERM_MCCTEMPBM_CRPT57490071___7]-	the NF service instance identified by the NF Service Instance ID and the NF instance Id (if available) or the last known NF instance ID.
(NOTE 2)
	NF-Service-Instance: serv1.smf1; NF-Inst: 54804518-4191-46b3-955c-ac631f953ed8

	NF-Service-Set
	NF Service Set ID
	All notifications (or callbacks) bound to:
[bookmark: _PERM_MCCTEMPBM_CRPT57490072___7]-	the NF Service Set ID; or
-	an NF service instance of the NF Service Set ID.
(NOTE 2)
	NF-Service-Set: setxyz.snnsmf-pdusession.nfi54804518-4191-46b3-955c-ac631f953ed8.5gc.mnc012.mcc345

	NF-Instance or NF-Set

and

Service-Name

	As defined above

and

as defined for servname in clause 5.2.3.2.5
	All notifications (or callbacks) bound to the service indicated in Service-Name within the NF instance ID or NF Set ID, as defined above

(NOTE 2)
	NF-Instance: 54804518-4191-46b3-955c-ac631f953ed8; Service-Name:def

	NOTE 1:	A notification (or callback) URI matches the Callback-Uri parameter value if the former contains the same scheme, the same authority and has a path that encompasses the path of the latter.
NOTE 2:	Notification (or callbacks) may be bound to an NF instance, an NF set, an NF service instance or an NF service set by a request creating a subscription or a callback resource with a Binding Indication as specified in clauses 6.12.4 and 6.12.5.



EXAMPLE 1:	Assuming that a PCF has created the following subscriptions in an AMF:
-	subscription 1: notification URI=https://pcf12.example.com/serviceX/1234
-	subscription 2: notification URI=https://pcf12.example.com/serviceY/abc
-	subscription 3: notification URI=https://pcf12.example.com/serviceY/def
When experiencing overload, if the PCF signals the following OCI scope:
-	Callback-Uri: ="https://pcf12.example.com", the OCI applies to notifications of all the subscriptions;
-	Callback-Uri: ="https://pcf12.example.com/serviceY", the OCI applies to notifications of subscriptions 2 and 3;
-	Callback-Uri: ="https://pcf12.example.com/serviceY/abc", the OCI applies to notifications of subscription 2.
EXAMPLE 2:	Assuming that a PCF has created the following subscriptions in an AMF:
-	subscription 1: notifications bound to PCF service set X, within PCF12 Instance ID, within PCF Set Z
-	subscription 2: notifications bound to PCF service set Y, within PCF12 Instance ID, within PCF Set Z
-	subscription 3: notifications bound to PCF12 Instance ID and service "def", within PCF Set Z
When experiencing overload, if the PCF signals the following OCI scope:
-	NF-Instance=PCF12 Instance ID, the OCI applies to notifications of all the subscriptions;
-	NF-Service-Set=Service Set Y of PCF12 Instance ID, the OCI applies to notifications of subscription 2;
-	NF-Instance=PCF12 Instance ID and Service="def", the OCI applies to notifications of subscription 3.

* * * Next Change * * * *
[bookmark: _Toc138412760]D.2	ABNF definitions (Filename: "TS29500_CustomHeaders.abnf")

; ----------------------------------------
;   RFC 5234
; ----------------------------------------

HTAB   = %x09 ; horizontal tab

LF     = %x0A ; linefeed

CR     = %x0D ; carriage return

SP     = %x20

DQUOTE = %x22 ; " (Double Quote)

DIGIT  = %x30-39 ; 0-9

ALPHA  = %x41-5A / %x61-7A ; A-Z / a-z

VCHAR  = %x21-7E ; visible (printing) characters

WSP    = SP / HTAB ; white space

CRLF   = CR LF ; Internet standard newline

HEXDIG = DIGIT / "A" / "B" / "C" / "D" / "E" / "F"



; ----------------------------------------
;   RFC 3986
; ----------------------------------------

unreserved    = ALPHA / DIGIT / "-" / "." / "_" / "~"

pct-encoded   = "%" HEXDIG HEXDIG

sub-delims    = "!" / "$" / "&" / "'" / "(" / ")" / "*" / "+" / "," / ";" / "="

pchar         = unreserved / pct-encoded / sub-delims / ":" / "@"

segment       = *pchar

segment-nz    = 1*pchar

path-abempty  = *( "/" segment )

path-absolute = "/" [ segment-nz *( "/" segment ) ]

path-rootless = segment-nz *( "/" segment )

path-empty    = 0pchar

IPvFuture     = "v" 1*HEXDIG "." 1*( unreserved / sub-delims / ":" )

dec-octet     = "25" %x30-35 / "2" %x30-34 DIGIT / "1" 2DIGIT / %x31-39 DIGIT / DIGIT

h16           = 1*4HEXDIG

ls32          = ( h16 ":" h16 ) / Ipv4address

Ipv4address   = dec-octet "." dec-octet "." dec-octet "." dec-octet

Ipv6address   = 6( h16 ":" ) ls32 
              / "::" 5( h16 ":" ) ls32 
              / [ h16 ] "::" 4( h16 ":" ) ls32 
              / [ *1( h16 ":" ) h16 ] "::" 3( h16 ":" ) ls32 
              / [ *2( h16 ":" ) h16 ] "::" 2( h16 ":" ) ls32
              / [ *3( h16 ":" ) h16 ] "::" h16 ":" ls32
              / [ *4( h16 ":" ) h16 ] "::" ls32
              / [ *5( h16 ":" ) h16 ] "::" h16
              / [ *6( h16 ":" ) h16 ] "::"

IP-literal    = "[" ( Ipv6address / IpvFuture  ) "]"

reg-name      = *( unreserved / pct-encoded / sub-delims )

host          = IP-literal / Ipv4address / reg-name

port          = *DIGIT

scheme        = ALPHA *( ALPHA / DIGIT / "+" / "-" / "." )

userinfo      = *( unreserved / pct-encoded / sub-delims / ":" )

authority     = [ userinfo "@" ] host [ ":" port ]

hier-part     = "//" authority path-abempty / path-absolute / path-rootless / path-empty

query         = *( pchar / "/" / "?" )

fragment      = *( pchar / "/" / "?" )

URI           = scheme ":" hier-part [ "?" query ] [ "#" fragment ]



; ----------------------------------------
;   RFC 5322
; ----------------------------------------

obs-FWS         = 1*WSP *( CRLF 1*WSP )

FWS             = ( [ *WSP CRLF ] 1*WSP ) / obs-FWS

obs-NO-WS-CTL   = %d1-8 / %d11 / %d12 / %d14-31 / %d127

obs-ctext       = obs-NO-WS-CTL

ctext           = %d33-39 / %d42-91 / %d93-126 / obs-ctext

obs-qp          = "\" ( %d0 / obs-NO-WS-CTL / LF / CR )

quoted-pair     = ( "\" ( VCHAR / WSP ) ) / obs-qp

ccontent        = ctext / quoted-pair / comment

comment         = "(" *( [ FWS ] ccontent ) [ FWS ] ")"

CFWS            = ( 1*( [ FWS ] comment ) [ FWS ] ) / FWS

day-name        = "Mon" / "Tue" / "Wed" / "Thu" / "Fri" / "Sat" / "Sun"

obs-day-of-week = [ CFWS ] day-name [ CFWS ]

day-of-week     = ( [ FWS ] day-name ) / obs-day-of-week

obs-day         = [ CFWS ] 1*2DIGIT [ CFWS ]

day             = ( [ FWS ] 1*2DIGIT FWS ) / obs-day

month           = "Jan" / "Feb" / "Mar" / "Apr" / "May" / "Jun"
                / "Jul" / "Aug" / "Sep" / "Oct" / "Nov" / "Dec"

obs-year        = [ CFWS ] 2*DIGIT [ CFWS ]

year            = ( FWS 4*DIGIT FWS ) / obs-year

date            = day month year

obs-hour        = [ CFWS ] 2DIGIT [ CFWS ]

hour            = obs-hour / 2DIGIT

obs-minute      = [ CFWS ] 2DIGIT [ CFWS ]

minute          = obs-minute / 2DIGIT

obs-second      = [ CFWS ] 2DIGIT [ CFWS ]

second          = obs-second / 2DIGIT

time-of-day     = hour ":" minute [ ":" second ]

obs-zone        = "UT" / "GMT" / "EST" / "EDT" / "CST" / "CDT" / "MST" / "MDT"
                / "PST" / "PDT" / %d65-73 / %d75-90 / %d97-105 / %d107-122

zone            = ( FWS ( "+" / "-" ) 4DIGIT ) / obs-zone

time            = time-of-day zone

date-time       = [ day-of-week "," ] date time [ CFWS ]



; ----------------------------------------
;   RFC 6749
; ----------------------------------------

NQCHAR = %x21 / %x23-5B / %x5D-7E



; ----------------------------------------
;   RFC 7230
; ----------------------------------------

OWS            = *( SP / HTAB )

RWS            = 1*( SP / HTAB )

[bookmark: _Hlk139019561]tchar          = "!" / "#" / "$" / "%" / "&" / "'" / "*" / "+" / "-"
               / "." / "^" / "_" / "`" / "|" / "~" / DIGIT / ALPHA

token          = 1*tchar

BWS            = OWS

quoted-string  = DQUOTE *( qdtext / quoted-pair ) DQUOTE

qdtext         = HTAB / SP /%x21 / %x23-5B / %x5D-7E / obs-text

obs-text       = %x80-FF



; ----------------------------------------
;   RFC 7231
; ----------------------------------------

date1          = day-rfc7231 SP month-rfc7231 SP year-rfc7231 ; e.g., 02 Jun 1982

day-rfc7231    = 2DIGIT

month-rfc7231  = %x4A.61.6E ; "Jan", case-sensitive
               / %x46.65.62 ; "Feb", case-sensitive
               / %x4D.61.72 ; "Mar", case-sensitive
               / %x41.70.72 ; "Apr", case-sensitive
               / %x4D.61.79 ; "May", case-sensitive
               / %x4A.75.6E ; "Jun", case-sensitive
               / %x4A.75.6C ; "Jul", case-sensitive
               / %x41.75.67 ; "Aug", case-sensitive
               / %x53.65.70 ; "Sep", case-sensitive
               / %x4F.63.74 ; "Oct", case-sensitive
               / %x4E.6F.76 ; "Nov", case-sensitive
               / %x44.65.63 ; "Dec", case-sensitive

year-rfc7231   = 4DIGIT

codings        = content-coding / "identity" / "*"

content-coding = token

weight         = OWS ";" OWS "q=" qvalue

qvalue         = ( "0" [ "." *3DIGIT ] ) / ( "1" [ "." *3"0" ] )



; ----------------------------------------
;   RFC 7235
; ----------------------------------------

credentials = auth-scheme [ 1*SP ( token68
            / [ ( "," / auth-param ) *( OWS "," [ OWS auth-param ] ) ] ) ]

auth-scheme = token

auth-param  = token BWS "=" BWS ( token / quoted-string )

token68     = 1*( ALPHA / DIGIT / "-" / "." / "_" / "~" / "+" / "/" ) *"="



; ----------------------------------------
;   3GPP TS 29.500
;
;   Version: 18.32.0 (June September 2023)
;
;   ©(c) 2023, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
; ----------------------------------------


;
; Header: 3gpp-Sbi-Message-Priority
;

Sbi-Message-Priority-Header = "3gpp-Sbi-Message-Priority:" OWS
                              ( "3" %x30-31 / %x31-32 DIGIT / DIGIT ) OWS



;
; Header: 3gpp-Sbi-Callback
;

Sbi-Callback-Header = "3gpp-Sbi-Callback:" OWS cbtype *1( ";" OWS "apiversion=" majorversion ) OWS

cbchar              = "-" / "_" / DIGIT / ALPHA

cbtype              = 1*cbchar

majorversion        = *DIGIT  



;
; Header: 3gpp-Sbi-Target-apiRoot
;

Sbi-Target-ApiRoot-Header = "3gpp-Sbi-Target-apiRoot:" OWS sbi-scheme "://" sbi-authority
                            [ prefix ] OWS

sbi-scheme                = "https" / "http"

sbi-authority             = host [ ":" port ]

prefix                    = path-absolute



;
; Header: 3gpp-Sbi-Routing-Binding
;

Sbi-Routing-Binding-Header = "3gpp-Sbi-Routing-Binding:" OWS "bl=" blvalue
                             1*( ";" OWS parameter ) OWS

blvalue                    = "nf-instance"
                           / "nf-set"
                           / "nfservice-instance"
                           / "nfservice-set"

parametername              = "nfinst"
                           / "nfset"
                           / "nfservinst"
                           / "nfserviceset"
                           / "servname"
                           / "backupamfinst"
                           / "backupnf"

parameter                  = parametername "=" token



;
; Header: 3gpp-Sbi-Binding
;

Sbi-Binding-Header = "3gpp-Sbi-Binding:" OWS binding-element *( OWS "," OWS binding-element ) OWS

binding-element    = "bl=" blvalue 1*( ";" OWS bh-parameter )
                     [ ";" OWS recoverytime ]
                     [ ";" OWS notif-receiver ]
                     [ ";" OWS "group=" groupvalue ]
                     [ 1*( ";" OWS groupparameter ) ]
                     [ ";" OWS "no-redundancy=" no-red-value ]

bh-parametername   = parametername / "scope"

bh-parameter       = bh-parametername "=" token

recoverytime       = "recoverytime=" OWS DQUOTE date-time DQUOTE

notif-receiver     = "nr=" URI

groupvalue         = "true" / "false"

groupparametername = "oldgroupid"
                   / "groupid"
                   / "uribase"
                   / "oldnfinst"
                   / "oldservset"
                   / "oldservinst"
                   / "guami"

groupparameter     = groupparametername "=" token

no-red-value       = "true" 



;
; Header: 3gpp-Sbi-Producer-Id
;

Sbi-Producer-Id-Header = "3gpp-Sbi-Producer-Id:" OWS "nfinst=" nfinst
                         [ OWS ";" OWS "nfservinst=" nfservinst ]
                         [ OWS ";" OWS "nfset=" nfset ]
                         [ OWS ";" OWS "nfserviceset=" nfserviceset ] OWS

nfinst                 = 8HEXDIG "-" 4HEXDIG "-" 4HEXDIG "-" 4HEXDIG "-" 12HEXDIG

nfservinst             = token

nfset                  = token

nfserviceset           = token



;
; Header: 3gpp-Sbi-Oci
;

Sbi-Oci-Header  = "3gpp-Sbi-Oci:" OWS oci-element *( OWS "," OWS oci-element ) OWS

oci-element     = timestamp ";" RWS validityPeriod ";" RWS olcMetric ";" RWS olcScope

timestamp       = "Timestamp:" RWS DQUOTE date-time DQUOTE

validityPeriod  = "Period-of-Validity:" RWS 1*DIGIT "s"

olcMetric       = "Overload-Reduction-Metric:" RWS ( "100" / %x31-39 DIGIT / DIGIT ) "%"

olcScope        = nfProducerScope / nfConsumerScope / scpScope / seppScope

nfProducerScope = (   ( "NF-Instance:" RWS nfinst ) 
                    / ( "NF-Set:" RWS nfset ) 
                    / ( "NF-Service-Instance:" RWS nfservinst [ ";" RWS "NF-Inst:" RWS nfinst ] )
                    / ( "NF-Service-Set:" RWS nfserviceset )
                  ) [ ";" RWS sNssaiList ";" RWS dnnList ]

nfConsumerScope = ( "NFC-Instance:" RWS nfinst [ ";" RWS "Service-Name:" RWS servname ] )
                / ( "NFC-Set:" RWS nfset [ ";" RWS "Service-Name:" RWS servname ])
                / ( "NFC-Service-Instance:" RWS nfservinst [";" RWS "NF-Inst:" RWS nfinst ] )
                / ( "NFC-Service-Set:" RWS nfserviceset )
                / ( "Callback-Uri:" RWS DQUOTE URI DQUOTE *( RWS "&" RWS DQUOTE URI DQUOTE ))

servname        = token

scpScope        = "SCP-FQDN:" RWS fqdn

seppScope       = "SEPP-FQDN:" RWS fqdn

fqdn            = token

dnnList         = "DNN:" RWS 1*tchar *( RWS "&" RWS 1*tchar )

sNssaiList      = "S-NSSAI:" RWS snssai *( RWS "&" RWS snssai )

snssai          = 1*tchar



;
; Header: 3gpp-Sbi-Lci
;

Sbi-Lci-Header    = "3gpp-Sbi-Lci:" OWS lc-element *( OWS "," OWS lc-element ) OWS

lc-element        = timestamp ";" RWS lcMetric ";" RWS lcScope

lcMetric          = "Load-Metric:" RWS ( "100" / %x31-39 DIGIT / DIGIT ) "%"

lcScope           = lcNfProducerScope / scpScope / seppScope

lcNfProducerScope = (   ( "NF-Instance:" RWS nfinst )
                      / ( "NF-Set:" RWS nfset)
                      / ( "NF-Service-Instance:" RWS nfservinst [ ";" RWS "NF-Inst:" RWS nfinst ] )
                      / ( "NF-Service-Set:" RWS nfserviceset)
                    ) [ ";" RWS sNssaiList ";" RWS dnnList ";" RWS relativeCapacity ]

relativeCapacity  = "Relative-Capacity:" RWS ( "100" / 1*2DIGIT ) "%"



;
; Header: 3gpp-Sbi-Client-Credentials
;

Sbi-Client-Credentials-Header = "3gpp-Sbi-Client-Credentials:" OWS jwt OWS

jwt                           = 1*b64urlchar "." 1*b64urlchar "." 1*b64urlchar

b64urlchar                    = ALPHA / DIGIT / "-" / "_"



;
; Header: 3gpp-Sbi-Nrf-Uri
;

Sbi-Nrf-Uri-Header = "3gpp-Sbi-Nrf-Uri:" OWS nrfUriParam *( OWS ";" OWS nrfUriParam ) OWS

nrfUriParam        = nrfUriParamName ":" RWS ( nrfUriParamValue1 / nrfUriParamValue2 )

nrfUriParamName    = "nnrf-disc" / "nnrf-nfm" / "nnrf-oauth2" / "oauth2-requested-services" / token

nrfUriParamValue1  = DQUOTE URI DQUOTE

nrfUriParamValue2  = ( nrfServiceName *( RWS "&" RWS nrfServiceName ) )

nrfServiceName     = "nnrf-disc" / "nnrf-nfm"



;
; Header: 3gpp-Sbi-Target-Nf-Id
;

Sbi-Target-Nf-Id-Header = "3gpp-Sbi-Target-Nf-Id:" OWS "nfinst=" nfinst
                          [ ";" OWS "nfservinst=" nfservinst ] OWS



;
; Header: 3gpp-Sbi-Max-Forward-Hops
;

Sbi-Max-Forward-Hops-Header = "3gpp-Sbi-Max-Forward-Hops:" OWS ( %x31-39 DIGIT / DIGIT )
                              ";" OWS "nodetype=" nodetypevalue OWS

nodetypevalue               = "scp"



;
; Header: 3gpp-Sbi-Originating-Network-Id
;

Sbi-Originating-Network-Id-Header = "3gpp-Sbi-Originating-Network-Id:" OWS 3DIGIT "-" 2*3DIGIT
                                    [ "-" 11HEXDIG ] [ ";" OWS srcinfo ] OWS

srcinfo                           = "src" ":" RWS srctype "-" srcfqdn

srctype                           = "SCP" / "SEPP"

srcfqdn                           = 4*( ALPHA / DIGIT / "-" / "." )



;
; Header: 3gpp-Sbi-Access-Scope
;

Sbi-Access-Scope-Header = "3gpp-Sbi-Access-Scope:" OWS scope-token *( SP scope-token ) OWS

scope-token             = 1*NQCHAR



;
; Header: 3gpp-Sbi-Access-Token
;

Sbi-Access-Token-Header = "3gpp-Sbi-Access-Token:" OWS credentials OWS



;
; Header: 3gpp-Sbi-Target-Nf-Group-Id
;

Sbi-Target-Nf-Group-Id-Header = "3gpp-Sbi-Target-Nf-Group-Id:" OWS "nfgid=" nfGroupIdValue OWS

nfGroupIdValue                = DQUOTE token DQUOTE



;
; Header: 3gpp-Sbi-Nrf-Uri-Callback
;

Sbi-Nrf-Uri-Callback-Header = "3gpp-Sbi-Nrf-Uri-Callback:" OWS
                              nrfUriCallbackParam *( OWS ";" OWS nrfUriCallbackParam ) OWS

nrfUriCallbackParam         = nrfUriCallbackParamName ":" RWS nrfUriCallbackParamValue

nrfUriCallbackParamName     = "nnrf-disc" / "nnrf-nfm" / token

nrfUriCallbackParamValue    = DQUOTE URI DQUOTE



;
; Header: 3gpp-Sbi-NF-Peer-Info
;

Sbi-NF-Peer-Info-Header = "3gpp-Sbi-NF-Peer-Info:" OWS peerinfo *( ";" OWS peerinfo ) OWS

peerinfo                = peertype "=" token

peertype                = "srcinst"
                        / "srcservinst"
                        / "srcscp"
                        / "srcsepp"
                        / "dstinst"
                        / "dstservinst"
                        / "dstscp"
                        / "dstsepp"



;
; Header: 3gpp-Sbi-Sender-Timestamp
;

Sbi-Sender-Timestamp-Header = "3gpp-Sbi-Sender-Timestamp:" OWS
                              day-name "," SP date1 SP time-of-day "." milliseconds SP "GMT" OWS

milliseconds                = 3DIGIT



;
; Header: 3gpp-Sbi-Max-Rsp-Time
;

Sbi-Max-Rsp-Time-Header = "3gpp-Sbi-Max-Rsp-Time:" OWS 1*5DIGIT OWS



;
; Header: 3gpp-Sbi-Correlation-Info
;

Sbi-Correlation-Info-Header = "3gpp-Sbi-Correlation-Info:" OWS
                              correlationinfo *( ";" OWS correlationinfo ) OWS

correlationinfo             = ctype "-" cvalue

ctype                       = "imsi" / "impi" / "suci" / "nai" / "gci" / "gli" / "impu"
                            / "msisdn" / "extid" / "imeisv" / "imei" / "mac" / "eui" / token

cvalue                      = 1*( tchar / "@" )



;
; Header: 3gpp-Sbi-Alternate-Chf-Id
;

Sbi-Alternate-Chf-Id-Header = "3gpp-Sbi-Alternate-Chf-Id:" OWS
                              "nfinst=" nfinst ";" OWS ( "primary" / "secondary" ) OWS



;
; Header: 3gpp-Sbi-Notif-Accepted-Encoding
;

Sbi-Notif-Accepted-Encoding-Header = "3gpp-Sbi-Notif-Accepted-Encoding:" OWS
                                     encoding-element *( OWS "," OWS encoding-element ) OWS

encoding-element                   = codings [ weight ]



;
; Header: 3gpp-Sbi-Consumer-Info
;

Sbi-Consumer-Info-Header = "3gpp-Sbi-Consumer-Info:" OWS
                           consumer-info-element *( OWS "," OWS consumer-info-element ) OWS

consumer-info-element    = ( supportedService ";" OWS supportedVersions 
                             [ ";" OWS supportedFeatures ]
                             [ ";" OWS acceptEncoding ]
                             [ ";" OWS callback-uri-prefix ]
                           ) [ ";" OWS intraPlmnCallbackRoot ";" OWS interPlmnCallbackRoot ]

supportedService         = "service=" servicename

servicename              = 1*( "-" / %x61-7A )

supportedVersions        = "apiversion=" "(" OWS
                           [ apimajorversion *( RWS apimajorversion ) OWS ] ")"

apimajorversion          = %x31-39 [ *DIGIT ]

supportedFeatures        = "supportedfeatures=" features

features                 = *HEXDIG

acceptEncoding           = "acceptencoding=" %x22 encodingList %x22

encodingList             = [ encoding-element *( OWS "," OWS encoding-element ) ]

intraPlmnCallbackRoot    = "intraPlmnCallbackRoot="
                           DQUOTE sbi-scheme "://" sbi-authority [ prefix ] DQUOTE

interPlmnCallbackRoot    = "interPlmnCallbackRoot="
                           DQUOTE sbi-scheme "://" sbi-authority [ prefix ] DQUOTE

callback-uri-prefix      = "callback-uri-prefix=" DQUOTE prefix DQUOTE



;
; Header: 3gpp-Sbi-Response-Info
;

Sbi-Response-Info-Header = "3gpp-Sbi-Response-Info:" OWS
                           resp-info-param *( OWS ";" OWS resp-info-param ) OWS

resp-info-param          = resp-info-param-name "=" OWS resp-info-param-value

resp-info-param-name     = "request-retransmitted"
                         / "nfinst"
                         / "nfset"
                         / "nfservinst"
                         / "nfserviceset"
                         / "context-transferred"
                         / "no-retry"
                         / token

resp-info-param-value    = token



;
; Header: 3gpp-Sbi-Selection-Info
;

Sbi-Selection-Info-Header = "3gpp-Sbi-Selection-Info:" OWS
                            selection-info-element *( OWS "," OWS selection-info-element ) OWS

selection-info-element    = ( "reselection=" reselectionvalue *( ";" OWS selection-criteria ) )
                          / ( selection-criteria *( ";" OWS selection-criteria ) )
reselectionvalue          = "true" / "false"

selection-criteria        = selection-action "=" token

selection-action          = "not-select-nfservinst"
                          / "not-select-nfserviceset"
                          / "not-select-nfinst"
                          / "not-select-nfset"



;
; Header: 3gpp-Sbi-Interplmn-Purpose
;

Sbi-Interplmn-Purpose-Header = "3gpp-Sbi-Interplmn-Purpose:" OWS N32Purpose ":" OWS
                               additional-info OWS

N32Purpose                   = "ROAMING"
                             / "INTER_PLMN_MOBILITY"
                             / "SMS_INTERCONNECT"
                             / "ROAMING_TEST"
                             / "INTER_PLMN_MOBILITY_TEST"
                             / "SMS_INTERCONNECT_TEST"
                             / "SNPN_INTERCONNECT"
                             / "SNPN_INTERCONNECT_TEST"
                             / "DISASTER_ROAMING"
                             / "DISASTER_ROAMING_TEST"
                             / token

additional-info              = token



;
; Header: 3gpp-Sbi-Request-Info
;

Sbi-Request-Info-Header = "3gpp-Sbi-Request-Info:" OWS req-param *( ";" OWS req-param ) OWS

req-param               = req-param-name "=" OWS req-param-value

req-param-name          = "retrans" 
                        / "redirect"
                        / "reason"
                        / "idempotency-key"
                        / "receivedrejectioncause"
                        / token

req-param-value         = token



;
; Header: 3gpp-Sbi-Retry-Info
;

Sbi-Retry-Info-Header = "3gpp-Sbi-Retry-Info:" OWS retriesindication OWS

retriesindication     = "no-retries"


* * * End of Changes * * * *




